NAG Library Routine Document

F01QGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F01QGF reduces the m by n ($m \leq n$) real upper trapezoidal matrix A to upper triangular form by means of orthogonal transformations.

2 Specification

```
SUBROUTINE F01QGF (M, N, A, LDA, ZETA, IFAIL)

INTEGER            M, N, LDA, IFAIL
REAL (KIND=nag_wp) A(LDA,*), ZETA(M)
```

3 Description

The m by n ($m \leq n$) real upper trapezoidal matrix A given by

$$A = \begin{pmatrix} U & X \end{pmatrix},$$

where U is an m by m upper triangular matrix, is factorized as

$$A = \begin{pmatrix} R & 0 \end{pmatrix} P^{\mathrm{T}},$$

where P is an n by n orthogonal matrix and R is an m by m upper triangular matrix.

P is given as a sequence of Householder transformation matrices

$$P = P_m \cdots P_2 P_1,$$

the $(m-k+1)$th transformation matrix, P_k, being used to introduce zeros into the kth row of A. P_k has the form

$$P_k = \begin{pmatrix} I & 0 \\ 0 & T_k \end{pmatrix},$$

where

$$T_k = I - u_k u_k^{\mathrm{T}},$$

$$u_k = \begin{pmatrix} \zeta_k \\ 0 \\ z_k \end{pmatrix},$$

ζ_k is a scalar and z_k is an $(n-m)$ element vector. ζ_k and z_k are chosen to annihilate the elements of the kth row of X.

The vector u_k is returned in the kth element of the array ZETA and in the kth row of A, such that ζ_k is in ZETA(k) and the elements of z_k are in A$(k, m+1), \ldots,$ A(k, n). The elements of R are returned in the upper triangular part of A.

For further information on this factorization and its use see Section 6.5 of Golub and Van Loan (1996).

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Wilkinson J H (1965) *The Algebraic Eigenvalue Problem* Oxford University Press, Oxford

5 Parameters

1: M – INTEGER *Input*

On entry: m, the number of rows of the matrix A.

When $\mathrm{M} = 0$ then an immediate return is effected.

Constraint: $\mathrm{M} \geq 0$.

2: N – INTEGER *Input*

On entry: n, the number of columns of the matrix A.

Constraint: $\mathrm{N} \geq \mathrm{M}$.

3: A(LDA,$*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the leading m by n upper trapezoidal part of the array A must contain the matrix to be factorized.

On exit: the m by m upper triangular part of A will contain the upper triangular matrix R, and the m by $(n-m)$ upper trapezoidal part of A will contain details of the factorization as described in Section 3.

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F01QGF is called.

Constraint: $\mathrm{LDA} \geq \max(1, \mathrm{M})$.

5: ZETA(M) – REAL (KIND=nag_wp) array *Output*

On exit: ZETA(k) contains the scalar ζ_k for the $(m-k+1)$th transformation. If $T_k = I$ then ZETA(k) = 0.0, otherwise ZETA(k) contains ζ_k as described in Section 3 and ζ_k is always in the range $\left(1.0, \sqrt{2.0}\right)$.

6: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: $\mathrm{IFAIL} = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= -1$

On entry, M < 0,
or N $<$ M,
or LDA $<$ M.

7 Accuracy

The computed factors R and P satisfy the relation

$$(R0)P^{\mathrm{T}} = A + E,$$

where

$$\|E\| \leq c\epsilon\|A\|,$$

ϵ is the ***machine precision*** (see X02AJF), c is a modest function of m and n and $\|.\|$ denotes the spectral (two) norm.

8 Further Comments

The approximate number of floating point operations is given by $2m^2(n-m)$.

9 Example

This example reduces the 3 by 5 matrix

$$A = \begin{pmatrix} 2.4 & 0.8 & -1.4 & 3.0 & -0.8 \\ 0.0 & 1.6 & 0.8 & 0.4 & -0.8 \\ 0.0 & 0.0 & 1.0 & 2.0 & 2.0 \end{pmatrix}$$

to upper triangular form.

9.1 Program Text

```
    Program f01qgfe

!     F01QGF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: f01qgf, nag_wp, x04cbf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter                :: indent = 0, ncols = 80, nin = 5,     &
                                           nout = 6
      Character (1), Parameter          :: diag = 'N', matrix = 'G', nolabel =  &
                                           'N'
      Character (4), Parameter          :: form = 'F8.4'
!     .. Local Scalars ..
      Integer                           :: i, ifail, lda, m, n
      Character (63)                    :: title
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable   :: a(:,:), zeta(:)
      Character (1)                     :: dummy(1)
!     .. Executable Statements ..
      Write (nout,*) 'F01QGF Example Program Results'
```

```
      Write (nout,*)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) m, n
      lda = m
      Allocate (a(lda,n),zeta(m))
      Read (nin,*)(a(i,1:n),i=1,m)

!     ifail: behaviour on error exit
!             =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
!     Find the RQ factorization of A
      Call f01qgf(m,n,a,lda,zeta,ifail)

      Write (nout,*) 'RQ factorization of A'
      Write (nout,*)
      Write (nout,*) 'Vector ZETA'
      Write (nout,99999) zeta(1:m)
      Write (nout,*)
      Flush (nout)
      title = &
        'Matrix A after factorization (R is in left-hand upper triangle)'
      ifail = 0
      Call x04cbf(matrix,diag,m,n,a,lda,form,title,nolabel,dummy,nolabel, &
        dummy,ncols,indent,ifail)

99999 Format (5(1X,F8.4))
    End Program f01qgfe
```

9.2 Program Data

```
F01QGF Example Program Data
  3     5                        : m, n
  2.4   0.8  -1.4   3.0  -0.8
   0    1.6   0.8   0.4  -0.8
   0     0    1.0   2.0   2.0   : a
```

9.3 Program Results

```
 F01QGF Example Program Results

 RQ factorization of A

 Vector ZETA
   1.2649   1.3416   1.1547

 Matrix A after factorization (R is in left-hand upper triangle)
   -4.0000 -1.0000 -1.0000  0.6325 -0.0000
    0.0000 -2.0000  0.0000  0.0000 -0.4472
    0.0000  0.0000 -3.0000  0.5774  0.5774
```